Pensieve header: Formal Gaussian integration over an arbitrary “Feynman Ring”.

What must a Feynman Ring F have? (Over some set of labels $\mathcal{S}$ with elements $x,\ y,\ \ldots$)
* A vector space over $\mathbb{Q}$.
* Has a symmetric linear $Z \longmapsto \partial_{x,y} Z$ and a symmetric bilinear $(Z_1, Z_2) \longmapsto (\partial_x Z_1)\left(\partial_y Z_2\right)$ that satisfy the axioms of (roughly) a connected circuit algebra.
* Has $q_{x,y} : F \to \mathbb{Q}$ in some sense dual to some $\theta_{x,y} \in F$.
* Has $\mathrm{Ev}_{\mathrm{vs}\to 0} : F \to F$.

Further axioms must be worked out.

Goals.
* Define $\int$.
* Prove a Fubini theorem.
* Prove a theorem about the injectivity of the Laplace transform.

Initialization

In[∘]:=
```
CCF[ℰ_] := ExpandDenominator@ExpandNumerator@Together[ℰ];
CCF[ℰ_] := Factor[ℰ];
CF[ω_. ℰ_𝔼] := CF[ω] CF /@ ℰ;
CF[ℰ_List] := CF /@ ℰ;
CF[ℰ_] := Module[{vs = Cases[ℰ, (x | p | π)__, ∞] ⋃ {x, p, ϵ, $A, $B, $C, $D}, ps, c},
   Total[CoefficientRules[Expand[ℰ], vs] /. (ps_ → c_) :→ CCF[c] (Times @@ vs^ps)]];
```

The Basic Feynman Ring

In[∘]:=
```
𝒮 = {x, x_, y, z};
q_{x_,y_}[f_] := (∂_{x,y} f) /. Thread[𝒮 → 0];
θ_{x_,y_} := x y;
f_ ≡ 0 := f === 0;
Ev_{vs_List→0}[f_] := CF[f /. Thread[vs → 0]]
```

The ϵSeries Feynman Ring

```
In[∘]:= 𝒮 = {x, y, z, ϕ, x_, p_, x̲_, p̲_};
q_{x_,y_}[ser_ϵSeries] := (∂_{x,y} ser⟦1⟧) /. Thread[𝒮 → 0];
θ_{x_,y_} := x y;
ϵSeries /: D[ser_ϵSeries, vs___] := D[#, vs] & /@ ser;
ϵSeries /: Plus[ss___ϵSeries] /; Length[{ss}] > 1 := Module[{l = Min[Length /@ {ss}]},
   ϵSeries @@ Total[Take[List @@ #, l] & /@ {ss}]]
ϵSeries /: t_ + ser_ϵSeries := MapAt[(# + t) &, ser, 1];
ϵSeries /: s1_ϵSeries * s2_ϵSeries :=
   ϵSeries @@ Table[Sum[s1⟦ii + 1⟧ s2⟦kk - ii + 1⟧, {ii, 0, kk}],
     {kk, 0, Min[Length@s1, Length@s2] - 1}] (*/.{A^2→0,B^2→0,A B→0}*);
ϵSeries /: c_ * ser_ϵSeries := ((c #) & /@ ser) (*/.{A^2→0,B^2→0,A B→0}*);
ser_ϵSeries ≡ 0 := And @@ ((# === 0) & /@ ser);
ϵSeries /: Integrate[ser_ϵSeries, pars__] := ϵSeries @@ (Integrate[#, pars] & /@ ser);
ϵSeries /: Ev_{vs_List→0}[ser_ϵSeries] := ser /. Thread[vs → 0];
CF[ser_ϵSeries] := (CF /@ ser) /. {$A^2 → 0, $B^2 → 0, $A $B → 0};
```

Integration

Using Picard Iteration!

```
In[∘]:= 𝔼 /: 𝔼[A_] 𝔼[B_] := 𝔼[A + B]
```

```
In[∘]:= 𝔼[sd_SeriesData] /; (List @@ sd)⟦{1, 2, 4, 6}⟧ === {ϵ, 0, 0, 1} :=
  𝔼[ϵSeries @@ PadRight[sd⟦3⟧, sd⟦5⟧, 0]]
```

pdf

Following a program in Projects/FullDoPeGDO/Engine.nb, we write $Z_\lambda = \sum Z[m]\,\lambda^m$.

In[∘]:=
```
Unprotect[Integrate];
Integrate::sing = "How dare you ask me to integrate a singular Gaussian!";
∫ω_. 𝔼[L_] ⅆ(vs_List) := Module[{n, Q, Δ, G, a, b, m, m1, $m}, Clear[Z];
   n = Length@vs;
   Q = Table[q_{vs⟦a⟧,vs⟦b⟧}[L], {a, n}, {b, n}];
   If[(Δ = CF@Det[-Q]) == 0, Message[Integrate::sing]; Return[]];
   G = CF[-Inverse[Q] / 2];
   Z[] = Z[0] = CF[L - Sum[Q⟦a, b⟧ θ_{vs⟦a⟧,vs⟦b⟧}, {a, n}, {b, n}] / 2];
   Z[m_, a_] := Z[m, a] = CF@D[Z[m], vs⟦a⟧];
   Z[m_, a_, b_] /; a ≤ b := Z[m, a, b] = CF@D[Z[m, a], vs⟦b⟧];
   Z[m_, a_, b_] /; a > b := Z[m, b, a];
   For[$m = m = 0, m ≤ 2 $m, ++m,
    Z[m + 1] = CF@Sum[Sum[If[G⟦a, b⟧ === 0, 0,
         G⟦a, b⟧/(m + 1) (Z[m, a, b] + Sum[Z[m1, a] Z[m - m1, b], {m1, 0, m}])],
        {a, n}], {b, n}];
    If[! (Z[m + 1] ≡ 0), $m = m + 1; Z[] += Z[m + 1]];
   ];
   PowerExpand@Factor[ω ((2 π)^(n/2) / Δ^(1/2))] 𝔼[CF[Ev_{vs→0}[Z[]]]]
  ];
Protect[Integrate];
```

In[∘]:=
```
Assuming[λ > 0, ∫_{-∞}^{∞} Exp[ⅈ λ x_1^2 / 2] ⅆx_1]
```

Out[∘]=
$$\frac{(1+i)\sqrt{\pi}}{\sqrt{\lambda}}$$

In[∘]:=
```
∫𝔼[ⅈ λ x_1^2 / 2] ⅆ{x_1}
```

Out[∘]=
$$\frac{(-1)^{1/4}\sqrt{2\pi}\,\mathbb{E}[0]}{\sqrt{\lambda}}$$

In[∘]:=
```
∫𝔼[-ⅈ λ x_1^2 / 2] ⅆ{x_1}
```

Out[∘]=
$$-\frac{(-1)^{3/4}\sqrt{2\pi}\,\mathbb{E}[0]}{\sqrt{\lambda}}$$

In[∘]:= $\int \mathbb{E}\left[\frac{i}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{2\pi\,\mathbb{E}[0]}{\sqrt{b^2 - a\,c}}$$

In[∘]:= $\int \mathbb{E}\left[-\lambda\, x_1^2 / 2\right] d\{x_1\}$

Out[∘]=

$$\frac{\sqrt{2\pi}\;\mathbb{E}[0]}{\sqrt{\lambda}}$$

In[∘]:= Clear[Z]; $\int \mathbb{E}\left[-x_1^2 / 2 + \xi\, x_1\right] d\{x_1\}$

Out[∘]=

$$\sqrt{2\pi}\;\mathbb{E}\left[\frac{\xi^2}{2}\right]$$

In[∘]:= $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1, x_2\}$

Out[∘]=

$$\frac{2\pi\,\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\left(-b^2 + a\,c\right)}\right]}{\sqrt{-b^2 + a\,c}}$$

In[∘]:= I1 = $\int \mathbb{E}\left[-\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c \end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\}\right] d\{x_1\}$

Out[∘]=

$$\frac{\sqrt{2\pi}\;\mathbb{E}\left[-\frac{\left(-b^2 + a\,c\right)\,x_2^2}{2\,a} + \frac{\xi_1^2}{2\,a} + \frac{x_2\,\left(-b\,\xi_1 + a\,\xi_2\right)}{a}\right]}{\sqrt{a}}$$

In[∘]:= $\int$ I1 $d\{x_2\}$

Out[∘]=

$$\frac{2\pi\,\mathbb{E}\left[\frac{c\,\xi_1^2 - 2\,b\,\xi_1\,\xi_2 + a\,\xi_2^2}{2\left(-b^2 + a\,c\right)}\right]}{\sqrt{-b^2 + a\,c}}$$

In[∘]:= $\int \mathbb{E}\left[\xi\, x + \eta\, y + z\,(x - y) + x^2\right] d\{x, z\}$

Out[∘]=

$$-2\,i\,\pi\,\mathbb{E}[y\,(y + \eta + \xi)]$$

Integration of ϵSeries

In[∘]:=
$$\int \mathbb{E}\left[-x^2/2 + \epsilon\, x^3/6 + O[\epsilon]^7\right] \, đ\{x\}$$

Out[∘]=
$$\sqrt{2\pi}\ \mathbb{E}\left[\epsilon\text{Series}\left[0,\ 0,\ \frac{5}{24},\ 0,\ \frac{5}{16},\ 0,\ \frac{1105}{1152}\right]\right]$$

In[∘]:=
$$\int \mathbb{E}\left[-\phi^2/2 + \epsilon\, \phi^4/24 + O[\epsilon]^7\right] \, đ\{\phi\}$$

Out[∘]=
$$\sqrt{2\pi}\ \mathbb{E}\left[\epsilon\text{Series}\left[0,\ \frac{1}{8},\ \frac{1}{12},\ \frac{11}{96},\ \frac{17}{72},\ \frac{619}{960},\ \frac{709}{324}\right]\right]$$

In[∘]:=
$$\int \mathbb{E}\left[p\, x + \epsilon\, p^2\, x + O[\epsilon]^5\right] \, đ\{p,\ x\}$$

Out[∘]=
$$-2\, i\, \pi\, \mathbb{E}[\epsilon\text{Series}[0,\ 0,\ 0,\ 0,\ 0]]$$

In[∘]:=
```
Block[{$π = Total@Select[MonomialList[#, {ϵ, x, p}],
      mon ↦ And[
        Exponent[mon, ϵ] ≤ 2,
        Exponent[mon, x] == Exponent[mon, p]
       ]
    ] &},
  ∫𝔼[p x + a x^2 p + ϵ b x^3 p^3] đ{p, x}]
```

Out[∘]=
$$-\frac{i\, \mathbb{E}\left[-6\, b\, \epsilon + 342\, b^2\, \epsilon^2\right]}{2\pi}$$

In[∘]:=
```
MatrixForm@Table[
  ∫𝔼[x₁ p₂ + x₂ p₃ + x₃ p₁ + ξᵢ xᵢ + πⱼ pⱼ] đ{x₁, x₂, x₃, p₁, p₂, p₃},
  {i, 3}, {j, 3}]
```

Out[∘]//MatrixForm=
$$\begin{pmatrix} -8\, i\, \pi^3\, \mathbb{E}[0] & -8\, i\, \pi^3\, \mathbb{E}[-\pi_2\, \xi_1] & -8\, i\, \pi^3\, \mathbb{E}[0] \\ -8\, i\, \pi^3\, \mathbb{E}[0] & -8\, i\, \pi^3\, \mathbb{E}[0] & -8\, i\, \pi^3\, \mathbb{E}[-\pi_3\, \xi_2] \\ -8\, i\, \pi^3\, \mathbb{E}[-\pi_1\, \xi_3] & -8\, i\, \pi^3\, \mathbb{E}[0] & -8\, i\, \pi^3\, \mathbb{E}[0] \end{pmatrix}$$